

## rangsor-kezelő függvények I.

ez a PDF dokumentum a [margitfalvi.hu](http://margitfalvi.hu)-n olvasható cikk nyomtatásra szánt változata

Ha egy szám-halmaz elemeit nagyság szerint sorrendbe állítjuk, úgy, hogy a legnagyobb szám álljon elől, akkor ez a halmaz elemeinek rangsora. Tessék csak nyugodtan úgy mondani Tanító bácsi, hogy torna-sora! Rendben, Pistike! Szóval, a rangsor a számok torna-sora. A halmaz egy elemének a rangsorban elfoglalt helyét, azaz rangsor-pozícióját, egy sorszámmal deklaráljuk.

A halmaz elemeinek számát a DARAB függvénnyel határozhatjuk meg. Ha a számok különbözőek, akkor az utolsó rangsor-pozíció azonos a halmaz elemeinek számával. A program rangsorkezelő függvényeivel közvetlenül meghatározhatjuk [1] a halmaz legnagyobb illetve legkisebb számát vagy értékét, [2] a rangsor elején illetve a rangsor végén, meghatározott helyen álló számot, [3] a rangsor közepén álló számot és [4] a halmaz elemeinek rangsor-pozícióját.

függvény	eredménye
MAX, MIN,	a halmaz legnagyobb-, legkisebb száma
MAXA, MIN2	kevert adattípusú halmaz legnagyobb-, legkisebb értéke
MAXHA, MINHA	szűkített halmaz legnagyobb-, legkisebb száma
NAGY, KICSI	a rangsor elején/végén, meghatározott helyen álló szám
MEDIÁN	a rangsor közepén álló szám
RANG.EGY, RANG.ÁTL	egy szám rangsor-pozíciója

A két MAX és a két MIN kötetlen argumentum-számú függvények. Természetesen ez nem végtelen számú argumentumot jelent, náluk is kettőszázötvenöt a határ. Az argumentumaik lehetnek konstansok, cella és tartomány hivatkozások és képletek. A MAX és a MIN függvények csak a számokat és a nem „hivatkozásként” megadott logikai értékeket vizsgálja. Az IGAZ értéke egy a HAMIS-é nulla.

	A	B	C	D
1	-5		6	
2	-2		9	
3	-4		5	
4	IGAZ		8	
5	-3		7	
6				
7	=MAX( A1:A5 ) » -2			
8	=MIN( C1:C5 ; HA( 1<2 ; IGAZ ) ) » 1			
9				

A két függvény a szövegeket a logikai értékkel azonos módon kezeli: ha „hivatkozásban” áll akkor figyelmen kívül hagyja, de képlet-argumentumban kiértékelésre kerül és hibát okoz (#ÉRTÉK!). A MAXA-val és MIN2-vel már mindkét adattípust korlátozás nélkül vizsgálhatjuk. A szöveg értéke nulla. Az üres cellák a négy függvény eredményét nem befolyásolják.

	A	B	C	D
1	-5		6	
2	-2		NA	
3	-4		5	
4	IGAZ		8	
5	-3		7	
6				
7	=MAXA( A1:A5 ) » 1			
8	=MIN2( C1:C5 ) » 0			
9				

Milyen bénaság ez, Tanító bácsi, hogy az egyik függvény-változatot A-val a másikat 2-sel jelölük? Én sejtem mi történhetett, Pistike! A jakutföldi Fegyva, aki a MAXA-t programozta és a kalkuttai Rahul,

aki a MIN2-t, valószínűleg nem tudhattak egymásról és redmondí Alex, akinek ellenőrizni kellett volna őket, gyengélkedett, mert elcsapta a gyomrát tegnap este fekete kaviárral.

Ha nem a teljes halmazt kívánjuk vizsgálni, akkor a MAXHA és a MINHA függvényeket használhatjuk. A függvényeknek azonos az argumentum listája.

MAXHA/MINHA( elemzendő tartomány ; feltétel tartomány<sub>1</sub> ; feltétel<sub>1</sub> ; feltétel tartomány<sub>2</sub> ; feltétel<sub>2</sub> ; ... ; feltétel tartomány<sub>127</sub> ; feltétel<sub>127</sub> )

A függvények első argumentuma tehát az a tartomány, ahol a legnagyobb illetve a legkisebb számot keresniük kell. További argumentumaik párban állnak. A párosok első eleme egy tartomány, amelynek celláiban a párosok második elemével deklarált feltételek teljesülését kell ellenőrizniük.

Az elemzendő tartománynak és a feltétel tartományoknak azonos méretűeknek és azonos tájolásúaknak kell lenniük. Az első három argumentum megadása kötelező. Az elemzendő tartomány üres és nem szám adattípusú elemei a függvények eredményét nem befolyásolják. A feltételek szintaktikája mindenben azonos a [feltételes megszámlálás függvényeinek](#) feltétel-argumentumaiban alkalmazandó szabályokkal.

A feltétel tartományok celláinak ellenőrzése sorrendben történik. Először a tartományok első celláit vizsgálja a függvény. Mindegyiket a saját feltételével. A következő munkamenetben a másodikcellák kerülnek sorra, majd minden harmadik. És így tovább, egészen az utolsó celláig. Ha az aktuálisan vizsgált cellák tartalma minden feltétel tartományban megfelelt a saját kritériumának, akkor az elemzendő tartomány azonos cellájában álló adat hozzáadódik a szűkített halmazhoz.

	A	B	C	D	E	F	G	H	I	J	K
1		létszám			darabszám			termelékenység			
2	részleg	jan.	feb.	már.	jan.	feb.	már.	jan.	feb.	már.	
3	A	17	13	11	17	34	27	80%	88%	82%	
4	B	15	10	17	34	21	17	71%	83%	72%	
5	C	14	19	10	24	33	20	73%	69%	84%	
6	D	11	20	14	34	15	29	78%	75%	81%	
7											
8		=MAXHA( H3:J6 ; B3:D6 ; "<=15" ; E3:G6 ; ">25" ) » 88%									
9											
10											

A fenti képen egy gyár termelési adatait látjuk. Arra voltunk kíváncsiak, hogy legfeljebb tizenötös létszám és legalább huszonhatos darabszám esetén mi volt a termelékenység legmagasabb értéke az első negyedévben.

A halmaz meghatározott sorszámú legnagyobb, illetve legkisebb elemét a kétargumentumos NAGY illetve KICSI függvényekkel választhatjuk ki. Például, a harmadik legnagyobbat vagy a második legkisebbet. Első argumentumukkal a halmazt tartalmazó tartományt deklaráljuk, a másodikkal pedig a kiíratandó elemet határozzuk meg a rangsor végpontjától kezdődő sorszámmal.

	A	B	C	D	E	F	G	H	I
1	97	-44 258	-6	257	-41	8	34 125	-579	
2									
3	=NAGY( A1:H1 ; 3 ) » 97				=KICSI( A1:H1 ; 2 ) » -579				
4									
5									
6	1	2	3	4	...				
7	34 125	257	97	8	-6	-41	-579	-44 258	
8				...	4	3	2	1	
9									
10									

A képen látható munkalap hetedik sorában a képzeletbeli rangsor áll, a balról illetve a jobbról induló sorszámokkal. Előbbieket a NAGY, utóbbiakat a KICSI függvény használja.

A két függvényt általában nem egyetlen-, hanem meghatározott számú elem kiírására használjuk. A következő képen erre a tipikus alkalmazásra látunk egy példát.

	L	M	N	O	P	Q	R	S
1	hallgató	pontszám						
2	Krisztián	149						
3	Sebestény	76						
4	Szaniszló	67	legmagasabb pontszámok					
5	Gyöngyvér	181		1	190	=NAGY( \$M\$2:\$M\$11 ; O4 )		
6	Zsuzsanna	94		2	181	=NAGY( \$M\$2:\$M\$11 ; O5 )		
7	Franciska	128		3	149	=NAGY( \$M\$2:\$M\$11 ; O6 )		
8	Magdaléna	90						
9	Annabella	78						
10	Annamária	125						
11	Krisztina	190						
12								

A függvény első argumentumának abszolút hivatkozása „rögzíti” az elemzendő tartományt a képlet másolásakor, míg a második argumentum relatív hivatkozása teszi lehetővé a sorszámok léptetését a másolatokban.

Szép, szép, Tanító bácsi! De az életben nem csak a legmagasabb pontszámokat akarjuk tudni, hanem azt is, hogy ki érte el őket! Így igaz Pistike! Ennek a feladatnak az elvégzésére az INDEX függvény szolgál... De használatát majd egy másik alkalommal mutatom be, most maradjunk a számok rangsoránál!

Ha valaki már hallott a mediánról, akkor biztos, hogy ezt hallotta: a középen álló szám. Meg azt, hogy: ugyan annyi van alatta, mint felette. Igen, a medián a rangsor közepén álló szám neve. Természetesen csak páratlan elem-számú halmaz esetén van ilyen pozíció. Ha nincs, akkor képezni kell egy nem valóságos-, egy ál-, egy pszeudó mediánt!

Az Excelben a rangsor közepén álló elemet a kötetlen argumentum-számú MEDIÁN függvénnyel választhatjuk ki. Argumentumai cella- és tartományhivatkozások, konstansok és képletek lehetnek. A függvény eredményét az üres cellák, a szövegek és a logikai értékek nem befolyásolják. Páros elem-szám esetén a képzett medián a rangsor közepén álló két szám számtani közepe.

	A	B	C	D	E
1	2	128	3		
2	=MEDIÁN( A1:C1 ) » 3				
3					
4					
5	2	128	3	0	
6	=MEDIÁN( A5:D5 ) » 2,5				
7					
8					

A képen az első sorban álló számok rangsora: 128, 3, 2. Középen a hármas áll, akkor ő lesz a három szám mediánja. Az ötödik sorban álló számok rangsora: 128, 3, 2, 0. A rangsor közepén a hármas és a kettes áll. Akkor ennek a két számnak a számtani közepe lesz a négy szám mediánja:  $(2+3)/2 \gg 2,5$ .

A MEDIÁN-t általában határérték képzésére használjuk: meghatározzuk vele azt a számot vagy azt a dátumot, amellyel egy csoport, elemeinek egy számszerűsíthető tulajdonsága alapján, két azonos nagyságú részre osztható. Vegyünk egy egyszerű példát: egy cégnél év végén maradt egy kis szétosztható pénz, de ez csak a kollégák felének elegendő. A főnök leginkább a hűségét díjazza. A belépési dátumok alapján határozzuk meg, kik számíthatnak a plusz pénzecskére!

	F	G	H	I	J	K	L	M
1	AZ	név	belépett					
2	001	Mikó Ábel	2017.01.01				HAMIS	
3	002	Goda Ernő	2013.06.06				IGAZ	1
4	003	Cseh Adél	2015.05.27				HAMIS	
5	004	Sebő Ádám	2014.11.06				HAMIS	
6	005	Róka Géza	2010.11.04				IGAZ	2
7	006	Rácz Iván	2011.11.01				IGAZ	3
8	007	Vári Hugó	2006.02.16				IGAZ	4
9	008	Sajó Áron	2010.04.10				IGAZ	5
10	009	Végh Benő	2017.04.04				HAMIS	
11	010	Simó Béla	2010.02.14				IGAZ	6
12	011	Fóti Márk	2007.04.04				IGAZ	7
13	012	Zeke Emil	2015.09.09				HAMIS	
14	013	Sütő Ákos	2016.08.30				HAMIS	
15	014	Engi Ottó	2014.01.20				HAMIS	
16	015	Duka Jenő	2018.09.07				HAMIS	
17	016	Vass Elek	2018.09.05				HAMIS	
18	017	Kósa Emőd	2008.04.03				IGAZ	8
19	018	Tóth Huba	2007.06.22				IGAZ	9
20								

Ahogy az azonosítókból (AZ) látjuk tizennyolc dolgozója van a cégnek. Miután megállapítottuk a belépések mediánját, egyenként megvizsgáltuk a dátumokat, kisebbek-e a mediánnál (=H2<\$j\$4). Kilenc ilyen dátumot találtunk...

Az eddig ismertetett függvények egy meghatározott rangsor-pozícióban álló számot adnak eredményül. A RANG.EGY ezzel szemben a halmaz egy számának rangsor-pozícióját állapítja meg. Három argumentuma van: az elsővel a számot, a másodikkal a halmazt tartalmazó tartományt, a harmadik, elhagyható argumentummal a sorszámozás irányát deklaráljuk. A harmadik argumentum lehetséges értékei: [hiányzik vagy 0] a legnagyobb számtól-, [1] a legkisebb számtól kezdődik a sorszámozás.

A függvény csak számokkal dolgozik, az üres cellákat, a szövegeket és a logikai értékeket figyelmen kívül hagyja. Ha az első argumentumával meghatározott számot a függvény nem találja a halmazban a #HIÁNYZIK hibaértéket kapjuk eredményül.

	A	B	C	D	E
1	-3	47	0	-25	
2					
3					
4					
5					
6					
7					

Az első sorban álló négy szám rangsora: 47, 0, -3, -25. A harmadik sor képletével a nulla rangsorpozícióját kívántuk megállapítani. Az eredmény kettő lett. A negyedik sor képlete csak a függvény harmadik argumentumával bővült: egy. Ez a deklaráció a sorszámozás megfordítására utasította a függvényt. Ezért kaptunk eredményül hármat. Az utolsó sorban álló képlettel az egy rangsor-pozícióját kívántuk megállapítani, de mivel a függvény az egyest nem találta a számok között, a #HIÁNYZIK hibaértéket kaptuk eredményül.

Jó, jó Tanító bácsi! De mi lesz az egyforma számokkal? Azok milyen sorszámot kapnak? Hát igen, Pistike! Ez megkerülhetetlen probléma. A legegyszerűbb ha nézünk egy példát és rögtön kipróbáljuk a RANG.ÁTL-ot is, mert annak az argumentum-listája azonosak a RANG.EGY-ével.

	A	B	C	D	E
1			RANG.EGY		RANG.ÁTL
2	-3		6		6
3	0		2		3,5
4	47		1		1
5	0		2		3,5
6	-25		7		7
7	0		2		3,5
8	0		2		3,5
9					
10			=RANG.EGY( A2 ; \$A\$2:\$A\$8 )		
11			=RANG.ÁTL( A2 ; \$A\$2:\$A\$8 )		
12					
13					

Ahogy látjuk a RANG.EGY „nem tesz” igazságot: az azonos számok a sorrendben következő pozíciószámot kapják. Az őket követő elem rangsor-pozíciója az utolsó „kiosztott” sorszám és az ismétlődések számának összege lesz. A példánkban, a rangsorban a négy nullát követő mínusz három rangsor-pozíciója:  $2+4 \gg 6$ . A RANG.ÁTL függvény csak az azonos számok sorszámozásában különbözik a testvérétől. Ő is azonos pozíció számot ad az ismétlődéseknek, de nem a következőt, hanem az ismétlődések által elfoglalt rangsor-pozíciók átlagát. Másként fogalmazva, az ismétlődések pozíció száma az elfoglalt rangsor-pozíciók száma összegének és az ismétlődések számának hányadosa. Tehát:  $(2+3+4+5)/4 \gg 3,5$ . Az ismétlődéseket követő szám rangsor-pozíciójának kiszámítása azonos az előbb ismertetett eljárással. Szándékosan vezettem be a „pozíció szám” kifejezést, hogy a „tényleges” rangsor-pozíciót ne mossam össze a függvény által visszaadott számmal.

Nekem egyik megoldás sem tetszik, Tanító Bácsi! Olyan béna ez a szakadozott sorszámozás! Szerintem is, Pistike! Ha sorszámozás, akkor legyen benne „sor”, vagyis folyamatosság! Van egy mesterfogás, amellyel megoldhatjuk a problémát. Azt mondjuk, hogy az az ismétlődő szám, amely előrébb áll a halmazban, vagyis a tartományban, az a rangsorban is megelőzi a halmazban öt követő ismétlődést. Tehát a RANG.EGY által visszaadott értékhez hozzá kell adni, a megszámlálás függvényeinél megismert, **előfordulás-számot**. Magyarul, ha már volt ilyen szám a halmazban, akkor azok darabszámát hozzá kell adni a pozíció-számhoz. Igen ám, de az előfordulás-szám tartalmazza az aktuálisan vizsgált értéket is! Ezért ebből ki kell vonnunk egyet.

	A	B	C	D	E	F	G	H
1	-3		6	=6+1-1		2	=2+1-1	
2	0		2	=2+1-1		3	=3+1-1	
3	47		1	=1+1-1		7	=7+1-1	
4	0		3	=2+2-1		4	=3+2-1	
5	-25		7	=7+1-1		1	=1+1-1	
6	0		4	=2+3-1		5	=3+3-1	
7	0		5	=2+4-1		6	=3+4-1	
8								
9				=RANG.EGY( A1 ; \$A\$1:\$A\$7 ) + DARABTELI( \$A\$1:A1 ; A1 ) - 1				
10				=RANG.EGY( A1 ; \$A\$1:\$A\$7 ; 1 ) + DARABTELI( \$A\$1:A1 ; A1 ) - 2				
11								
12								

Az A oszlopban álló számok rangsora: 47, 0 (A2), 0 (A4), 0 (A6), 0 (A7), -3, -25. Már bocsánat, Tanító bácsi! De kicsit röhejes, hogy megkülönböztetjük a nullákat! Nem tök mindegy! Hááát igen, Pistike! Elég nevetséges, de ha úgy gondolunk ezekre a számokra, mint egy csoport tagjainak az egyik számszerűsíthető tulajdonságára, akkor már nem mindegy, ki az ötödik és ki a hatodik, mert biztos nem véletlenül rangsorolták ezeket az embereket...

Itt a vége, fuss el véle! Na és a SORSZÁM-mal mi lesz? Jaj, tényleg, róla teljesen megfeledkeztem! Tudod, Pistike, amikor még nem volt RANG.EGY és még RANG.ÁTL se, akkor csak a SORSZÁM volt, ami pont ugyan úgy működött mint, a RANG.EGY. De, amikor már volt RANG.EGY és RANG.ÁTL is, a

SORSZÁM akkor is maradt. Állítólag a kompatibilitás miatt. Szerintem, ebben az esetben is Fegyva, Rahul és Alex voltak a szereplők, csak most nem a fekete kaviár volt sok, hanem a fehér koko.



[margitfalvi.arpad@gmail.com](mailto:margitfalvi.arpad@gmail.com)